
ABSTRACT 

If you share the results of your SAS
®
 programs with anyone;  you likely use SAS formats.   Each time you run your 

program, you recreate the formats.  This format recreation takes time, it adds to the length of your log, the formats do 
not change once you have created them and SAS issues warnings that the formats already exist.  Nonetheless, you 
still create them over and over again.  Well, did you know that you can easily create your formats once, save them 
and then re-use them again and again?  If you are feeling generous, you can even share them with others. 

INTRODUCTION 

SAS formats or the more generic term “value labels” are used to make coded information more user-friendly.  Instead 
of using codes 1 and 2 in your reports and charts, a meaningful label such as MALE and FEMALE or YES and NO 
takes the place of the code.  Typically, value labels are created at the beginning of a SAS program and applied to a 
SAS dataset as it is created.   Each time the program is run; the value labels are re-created and applied to the 
dataset. This takes time and adds to the length of your program and logs.  Each subsequent time that you run your 
program, SAS issues a warning that the formats already exist.  There is a solution.   SAS allows you to save your 
formats in a catalog so you can re-use them and, if you are feeling generous, share them with others.  

WHAT IS A SAS FORMAT? 

A SAS format is a user-friendly way of displaying numeric or character data.  Formats allow you to add meaning to 
codes, group codes or data points.  Formats also let you recode and reformat data.  Without explicit value labels, 
other users must rely on a codebook or data dictionary to interpret the data, the tables or the charts with which they 
are working.   You can create formats using the FORMAT procedure and you can use the FORMAT statement in a 
data step to connect a variable to a format.  Please see the SAS Online Documentation for more details.     

HOW TO SAVE A FORMAT 

All formats are stored in a catalog.  A catalog is just another term for a collection of formats.   Any time you create or 
use SAS formats, they get stored in a catalog.  If you don’t specify the catalog, SAS places your formats in the 
WORK library in a catalog called FORMATS.   These are temporary formats catalogs.  As with everything else in the 
WORK library, they are deleted at the end of each session. 

To save your SAS formats, you must tell SAS where to put the catalog and what to call it.  This is done by telling SAS 
to put your formats in a library other than the default WORK library.   So what is a SAS library?  A SAS library is just a 
"short cut" or a nickname to a subdirectory found somewhere on a hard drive.  You can use a LIBNAME statement to 
define the nickname. 

 LIBNAME SASDATA 'C:\SASDatasets' 

 The catalog can be called any valid dataset name. 

 PROC FORMAT LIBRARY=SASDATA.MYFILENAME; 

If you don’t specify a file name, SAS will provide one for you.  The default name is "FORMATS".  If the formats are for 
a specific file, it may be helpful to use the name of the SAS dataset file. 

 Now that you have a SAS format catalog, you have to tell SAS where that catalog is located and in what order to 
search through any catalogs you may have created. 

 OPTIONS FMTSEARCH = ( SASDATA.MYFILENAME ); 

The OPTIONS statement parameter, FMTSEARCH, tells SAS where format catalogs are located and in what order to 
look through those catalogs.  By default, the WORK.FORMATS catalog is always searched first unless you specify a 
different order. 

OPTIONS FMTSEARCH=( SASDATA.MYFILENAME WORK ); 

For convenience, SAS includes a library called LIBRARY in the search path.  If you use a LIBNAME statement to 
define a library called LIBRARY, then you do not have to use the OPTIONS FMTSEARCH statement.   If you are 
running on a server using the BI Platform, then you have an additional library called APFMTLIB.  The APFMTLIB 
library is common to everyone using that BI server. 

Note:  If you use the Enterprise Guide 4.3 tasks “Create Format” or “Create Format from Data Set” the APFMTLIB 
does not appear in the “Location to store format Library:” pick list.  A work around in order to have the library appear 
in the pick lists is to use PROC DATASETS LIBRARY=APFMTLIB ; RUN;  to wake-up the library. This is only 
required in the project that creates the formats. This should be corrected in Enterprise Guide 5.2. 

HOW TO USE SAVED FORMATS 

To use the saved formats in a new program, you have to define the library again with the LIBNAME statement and tell 
SAS to use the format catalog with the OPTIONS FMTSEARCH=( SASDATA.MYFILENAME ). 

 HOW TO TURN FORMATS ON/OFF 

Once a SAS dataset has a format associated with its variables,  SAS will always look for that format catalog which 
contains the listed formats when it opens the dataset.   If it does not find a catalog entry, it will issue an error 
message and a note.   SAS allows you to override this with the OPTIONS statement.    

 OPTIONS FMTSEARCH=( WORK LIBRARY) NOFMTERR NONOTES; 

In earlier versions of SAS, before SAS 9.3, the following statement would clear the search path to all the possible 
format libraries. 

  OPTIONS FMTSEARCH=() NOFMTERR NONOTES; 

SAS 9.3 produces an error message using this syntax. 

HOW TO DELETE FORMATS FROM A SAS DATASET 

Using the FORMAT statement in a data step, you leave the format name parameter blank.  By doing this, you can 
delete one, many or all the formats.  The new data set will not have any formats assigned. 

 FORMAT Q001; 
 FORMAT Q001-Q003 Q005; 
 FORMAT _ALL_; 

TIPS ON USING FORMATS 

In a SAS dataset, the format name is part of the definition of each of the variables.  There is no connection to the 
catalog other than the format name.  This means you can have more than one format for each variable.  

For example, you can switch languages by changing format catalogs.     

 OPTIONS FMTSEARCH=( SASDATA.ENGLISHFMT ); 

 OPTIONS FMTSEARCH=( SASDATA.FRENCHFMT ); 

To determine what libraries are available and in what order they will be searched use the OPTIONS procedure. 

  PROC OPTIONS OPTION=FMTSEARCH; RUN;  

This command lists the order of the format catalogs searched in the log.  If you have changed the order of the default 
catalogs, SAS will list the new order.   In other words, if you have added a catalog to the end of the list, SAS will not 
show you the whole list.  SAS will only show you the part that you have added.  If you have added your catalog to the 
beginning of the list then SAS will show the entire list and your new catalog will be shown first. 

SAMPLE PROGRAM 

** Before running the sample code create a subdirectory on your C: drive; 

 

** Create a library to store the sample data and format cataloges; 

LIBNAME SASDATA 'C:\SASDATA'; 

 

** Print the current format search string in the log; 

PROC OPTIONS OPTION=FMTSEARCH; 

RUN; 

 

** Create the English value labels; 

PROC FORMAT LIBRARY=SASDATA.English; 

    VALUE PROVFMT 

      10='Newfoundland and Labrador' 

      11='Prince Edward Island' 

      12='Nova Scotia' 

      13='New Brunswick' 

      24='Quebec' 

      35='Ontario' 

      46='Manitoba' 

      47='Saskatchewan' 

      48='Alberta' 

      59='British Columbia'; 

 

   VALUE YESNOFMT 

      1 = 'YES' 

      2 = 'NO' 

  OTHER = 'N/A'; 

RUN; 

 

** Create the French value labels; 

PROC FORMAT LIBRARY=SASDATA.French; 

    VALUE PROVFMT 

      10='Terre-Neuve et Labrador' 

      11="l'île du Prince-Édouard" 

      12='Nouvelle-Écosse' 

      13='Nouveau-Brunswick' 

      24='Québec' 

      35='Ontario' 

      46='Manitoba' 

      47='Saskatchewan' 

      48='Alberta' 

      59='Colombie britannique'; 

 

   VALUE YESNOFMT 

      1 = 'Qui' 

      2 = 'Non' 

  OTHER = 'N/A'; 

RUN; 

 

OPTIONS FMTSEARCH=( SASDATA.English ); 

 

** Create sample data; 

DATA SASDATA.THEDATA; 

    INPUT @01 ID   $02.  

          @03 Q01    1. 

          @04 Q02    1. 

          @05 Q03    1. 

          @06 WTS    4.2 

          @10 PROV   2.; 

 

   LABEL ID = 'Unique Identifier' 

        Q01 = 'Question One' 

        Q02 = 'Question Two' 

        Q03 = 'Question Three' 

        WTS = 'Weights' 

       PROV = 'Province'; 

   FORMAT Q01-Q03 YESNOFMT. PROV PROVFMT.; 

 

DATALINES4; 

011210.2310 

022120.3424 

031210.4547 

042120.1048 

051210.2910 

062120.3835 

071220.4759 

082100.5635 

091210.6510 

102010.7412 

112210.8311 

122220.9213 

132220.0147 

142120.1248 

150000.2135 

;;;; 

RUN; 

 

TITLE 'English Formats'; 

** Set the format search list; 

OPTIONS FMTSEARCH=( SASDATA.English ); 

 

** Print the current format search string in the log; 

PROC OPTIONS OPTION=FMTSEARCH; 

RUN; 

 

** Sample output of English value labels; 

PROC PRINT DATA=SASDATA.THEDATA; 

RUN; 

 

PROC FREQ DATA=SASDATA.THEDATA; 

   TABLES _ALL_ /MISSING; 

RUN; 

 

TITLE 'French Formats'; 

** Set the format search list; 

OPTIONS FMTSEARCH=( SASDATA.French ); 

** Print the current format search string in the log; 

PROC OPTIONS OPTION=FMTSEARCH; 

RUN; 

 

** Sample output of French value labels; 

PROC PRINT DATA=SASDATA.THEDATA; 

RUN; 

 

PROC FREQ DATA=SASDATA.THEDATA; 

   TABLES _ALL_ /MISSING; 

RUN; 

 

TITLE 'No Formats'; 

** Set the format search list; 

OPTIONS FMTSEARCH=( work ) nofmterr; 

** Print the current format search string in the log; 

PROC OPTIONS OPTION=FMTSEARCH; 

RUN; 

 

** Sample output of French value labels; 

PROC PRINT DATA=SASDATA.THEDATA; 

RUN; 

 

PROC FREQ DATA=SASDATA.THEDATA; 

   TABLES _ALL_ /MISSING; 

RUN; 

 

TITLE; 

 

** Delete the sample data and formats; 

PROC DATASETS LIBRARY=SASDATA; 

   DELETE THEDATA (memtype=data); 

   DELETE English (memtype=catalog); 

   DELETE French  (memtype=catalog); 

QUIT; 

CONTACT INFORMATION 

Your comments and questions are valued and encouraged. Contact the author at: 

Name:   John Ladds 
Enterprise:   Statistics Canada 
Address:   R.H. Coats Building, 14th Floor, Section R 90, 100 Tunney's Pasture Driveway 
City, State ZIP:  Ottawa, Ontario, Canada  
Work Phone:   613 951-1767 
Fax:   613 951-0607 
E-mail:   John.Ladds@statcan.gc.ca 
Web:    http://www.statcan.ca  
 

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS 
Institute Inc. in the USA and other countries. ® indicates USA registration.  

Other brand and product names are trademarks of their respective companies.  

